

Intégration d'une équation différentielle ordinaire

January 26, 2012

1 Position du problème

Pour décrire un processus d'évolution, ou le profil d'une ligne d'eau, par exemple, on est souvent amené à résoudre une équation différentielle ordinaire (EDO) du premier ordre. Cette équation écrit comment varie une fonction, en un point donné (un instant ou un point de l'espace), connaissant la valeur de cette fonction. Mathématiquement, le problème à résoudre s'écrit:

$$\begin{cases} \frac{dy}{dt} = f(y, t) \\ y(t = t_0) = y_0 \end{cases} \quad (1)$$

où $\frac{dy}{dt}$ désigne la dérivée par rapport à t de la fonction y (qui dépend de la variable t); la variable y_0 est appelée la condition à la limite; elle conditionne la solution finale de l'équation.

Comme souvent on ne connaît pas de solution analytique de ce problème, on va utiliser des méthodes approchées pour estimer la solution. On fait donc une discrétisation de la variable t . On note ainsi Δt le pas de discrétisation, et on résout le problème aux points $t_0, t_1 = t_0 + \Delta t, t_2 = t_0 + 2\Delta t, \dots, t_n = t_0 + n\Delta t$ où n est un entier.

2 Méthode d'Euler explicite

Cette méthode est la plus intuitive; elle consiste à considérer que, d'un point t_i au point t_{i+1} , la fonction évolue linéairement, avec une trajectoire qui est celle qu'on peut calculer au point t_i .

Le problème se résout donc de la façon suivante:

- on connaît la fonction f , un point t_i où on connaît y_i
- on peut donc calculer $y'_i = f(y, t)$
- on estime alors la valeur de y au point $t_{i+1} = t_i + \Delta t$:

$$y_{i+1} \simeq y_i + y'_i \Delta t$$

- on peut alors itérer (résoudre pas à pas) pour passer au point suivant. Le problème est initialisé en partant de t_0 où on connaît y_0 (condition à la limite).

On sent bien que ce schéma pourra donner de bons résultats uniquement si Δt n'est pas trop grand. Des valeurs de Δt trop grandes peuvent donner des résultats complètement faux, conduisant à des interprétations physiques erronées. Son intérêt est toutefois sa simplicité, et il s'implémente facilement sur un tableau.

Exemple d'application: processus exponentiel. Considérons le problème (simple) suivant:

$$\begin{cases} \frac{dy}{dt} = -ay \\ y(t = t_0) = y_0 \end{cases} \quad (2)$$

On a donc ici $f(y, t) = -ay$. La solution analytique se résout facilement, donnant $y(t) = y_0 \exp(-a(t - t_0))$. On peut résoudre le problème par la méthode d'Euler:

- on choisit Δt (par exemple, $\Delta t = 1$)
- calculer $y_1 = y_0 - ay_0 \Delta t$
- calculer $y_2 = y_1 - ay_1 \Delta t$ etc.

La résolution est présentée Fig. 1. On constate que la résolution n'est pas très précise; ceci est lié au pas de calcul trop grand compte tenu de la méthode choisie et de l'équation à résoudre.




Figure 1: Résolution de l'Eq. (2) par la méthode d'Euler

3 Schéma de Runge-Kutta d'ordre 4

Le schéma de Runge-Kutta d'ordre 4 est basée sur une approximation de la dérivée à un ordre supérieure (ordre 4). Le principe de la discrétisation est le même, mais on va faire quelques calculs supplémentaire pour approcher la dérivée:

- on connaît la fonction f , un point t_i où on connaît y_i
- on peut donc calculer $f_1 = y'_i = f(y_i, t_i)$ (cf. méthode d'Euler=pente au point (t_i, y_i))

- on calcule $f_2 = f(y_i + \frac{1}{2}\Delta t f_1, t_i + \frac{1}{2}\Delta t)$ (valeur estimée au milieu de l'intervalle, avec la pente prise en t_i)
- on calcule $f_3 = f(y_i + \frac{1}{2}\Delta t f_2, t_i + \frac{1}{2}\Delta t)$ (valeur estimée au milieu de l'intervalle $t_{i+1/2}$, avec la pente prise en $t_{i+1/2}$)
- on calcule $f_4 = f(y_i + \Delta t f_3, t_i + \Delta t)$ (valeur estimée en t_{i+1} , avec la pente prise en $t_{i+1/2}$)

- on a alors

$$y_{i+1} \simeq y_i + \frac{1}{6}\Delta t(f_1 + 2f_2 + 2f_3 + f_4)$$

- on peut alors itérer (résoudre pas à pas) pour passer au point suivant. Le problème est initialisé en partant de t_0 où on connaît y_0 (condition à la limite).

On voit clairement (Fig. 2) que la méthode est beaucoup plus précise. Même avec un pas de calcul beaucoup plus élevé, on approche correctement la solution, alors que la méthode d'Euler donne des résultats très éloignés de la solution exacte. On remarque aussi que, entre les points de discrétisation, on a approché la solution par des segments de droite (bien qu'on puisse avoir une interpolation plus évoluée si on a besoin de connaître des valeurs intermédiaires).




Figure 2: Résolution de l'équation différentielle, $\Delta t = 5$, avec la méthode d'Euler et la méthode de Runge-Kutta (RK4)

Script

Le programme peut s'écrire de la façon suivante, avec le logiciel de calcul scientifique SCilab:

```
// Programme de resolution d'une equation differentielle
// Equation a resoudre: dy/dt=f(y,t)

// Definition de la fonction f
a=-0.1;
function z=f(y,t);z=a*y; endfunction
```

```

// Condition à la limite:
t0=0;
y0=1;

// Discretisation du temps
tmax=50;
dt=5;

// Nombre de pas de discretisation
N=(tmax-t0)/dt;
// Indices
ii=1:N+1;
t=(ii-1)*dt; // vecteur temps

t2=0:tmax; // vecteur temps avec un pas plus fin, pour la solution exacte

// Solution par méthode d'Euler, notée ye
ye(1)=y0; // condition à la limite
for i=1:N
    ye(i+1)=ye(i)+dt*f(ye(i),t(i));
end

// Solution par méthode RK4, notée yrk4
yrk4(1)=y0; // condition à la limite
for i=1:N
    f1=f(yrk4(i),t(i));
    f2=f(yrk4(i)+f1*dt/2,t(i)+dt/2);
    f3=f(yrk4(i)+f2*dt/2,t(i)+dt/2);
    f4=f(yrk4(i)+f3*dt,t(i)+dt);
    yrk4(i+1)=yrk4(i)+dt*(f1+2*f2+2*f3+f4)/6;
end

// Tracé des solutions
scf(2)
plot(t2,exp(a*t2),'k-',t, ye,'kd-',t, yrk4,'ks-')
title('Resolution par méthode de Runge-Kutta d''ordre 4 - dy/dt=-0.1y')
xlabel('t')
ylabel('y')
legend('Solution exacte','Méthode Euler, dt=5','Méthode RK4, dt=5')

```

Exemple d'application: calcul d'une ligne d'eau On considère l'équation différentielle suivante:

$$\frac{dy}{dx} = \frac{I_f - J(h)}{1 - F^2(h)} \quad (3)$$

où I_f est la pente d'un canal, J la formule nous donnant la perte de charge locale (dépendant la hauteur d'eau), y désigne ici la hauteur d'eau. On donne ainsi, pour un canal rectangulaire de largeur b et coefficient de Strickler K :

$$J = \frac{Q^2(b + 2y)^{4/3}}{K^2 b^{10/3} y^{10/3}} \quad (4)$$

et

$$F^2 = \frac{Q^2}{gb^2y^3} \quad (5)$$

Il s'agit d'un cours d'eau en pente faible. On donne une condition à la limite à l'aval (y_0 , 0.12m dans notre exemple). Le code est modifié ainsi:

```
b=0.3;  
K=50;  
If=0.005;  
Q=0.01;  
function z=DQ(y);z=Q-K*(b*y)^(5/3)/(b+2*y)^(2/3)*sqrt(If); endfunction  
yn=fsolve(0.5,DQ);  
tmax=0;  
t0=10;  
dt=-0.5;  
function z=f(y,t);z=(If-Q^2*(b+2*y)^(4/3)/(K^2*(b*y)^(10/3)))/(1-Q^2/(9.81*b^2*y^3)); endfunction  
y0=0.12;
```

ce qui nous donne la profondeur normal $y_n=0.071$ m, et le tracé de la ligne d'eau (Fig. 3). Dans ce cas, les deux méthodes donnent des résultats proches. En revanche, on peut avoir de grosses erreurs dans le cas d'une courbe de remous de type 2 (condition aval sous la hauteur normale), car les pentes de ligne d'eau sont beaucoup plus fortes, et donc beaucoup plus sujettes aux erreurs liées à l'interpolation linéaire. On en déduit donc que d'une part le choix de la méthode de résolution est importante, d'autre part qu'il est essentiel d'avoir un regard critique sur les solutions (avec une interprétation des processus qu'on cherche à modéliser).




Figure 3: Résolution de l'équation de la ligne d'eau